計算機在工程應用— Gaussian Integration

$$\int_{-1}^{1} f(x)dx = ?$$

Motivation:

$$\int_{-1}^{1} f(x)dx = \sum_{k=0}^{n} w_i f(x_i)$$

where w_i is weighting and x_i is sampling point.

Special case

Trapezoidal rule:

$$n = 2, x_0 = -1, x_1 = 1, w_0 = 1, w_1 = 1$$

Simpson rule :

$$n = 3, x_0 = -1, x_1 = 0, x_2 = 1, w_0 = 1/6, w_1 = 4/6, w_2 = 1/6$$

Gaussian question ?

Why equal space ?

Derivation:

$$f_{2n+1}(x) = \sum_{i=0}^{n}[P_{n+1}(x_i)g_n(x_i) + h_n(x_i)]L_i^n(x)$$

First approximation for $f_{2n+1}(x) \to P_{n+1}(x)g_n(x)$

Remainder term $\to h_n(x)$

$$\begin{aligned} f_{2n+1}(x) &= P_{n+1}(x)g_n(x) + h_n(x) \\ &= \sum_{i=0}^{n}[P_{n+1}(x_i)g_n(x_i) + h_n(x_i)]L_i^n(x) \\ &= \sum_{i=0}^{n} f_{2n+1}(x_i)L_i^n(x) \end{aligned}$$

Note that $g_n(x)$ can be expanded in term of

$$g_n(x) = a_0P_0(x) + a_1P_1(x) + a_2P_2(x) + \cdots + a_nP_n(x)$$

Therefore,

$$\int_{-1}^{1} P_{n+1}(x) g_n(x) dx = 0$$

Therefore,

$$\begin{aligned}
\int_{-1}^{1} f_{2n+1}(x) dx &= \sum_{i=0}^{n} \int_{-1}^{1} h_n(x_i) L_i^n(x) dx \\
&= \sum_{i=0}^{n} h_n(x_i) \int_{-1}^{1} L_i^n(x) dx \\
&= \sum_{i=0}^{n} h_n(x_i) w_i \\
&= \sum_{i=0}^{n} f_{2n+1}(x_i) w_i
\end{aligned}$$

where x_i are the $(n+1)$ roots of $P_{n+1}(x)$ and $w_i = \int_{-1}^{1} L_i^n(x) dx$.

Relation between Langrange and Lengendre polynomials:

$$L_i^n(x) = \frac{1}{P'_{n+1}(x_i)} \frac{P_{n+1}(x)}{(x - x_i)}$$

check:

$$L_i^n(x_j) = 1, \;\; if \;\; j = i$$

$$L_i^n(x_j) = 0, \;\; if \;\; j \neq i$$

Example:

For $n = 2$,

$$P_3(x) = \frac{1}{2}(5x^3 - 3x)$$

$$x_0 = \sqrt{-3/5}, x_1 = 0, x_2 = \sqrt{3/5}$$

$$P_3'(x) = \frac{(15x^2 - 3)}{2}$$

$$L_0^2(x) = \frac{x(x - \sqrt{3/5})}{3(x_0^2 - \frac{1}{5})}$$

$$w_0 = 5/9, w_1 = 8/9, w_2 = 5/9$$

Homework:

1. Proof of $\sum_{i=0}^{n} w_i = 2$

Program Listing:

```
C*******************************************************
C         SUBROUTINE GRULE                             *
C   THIS IS A SUBROUTINE TO COMPUTE THE (N+1)/2        *
C   NONNEGATIVE AB-SCISSAS X(I),W(I) OF THE N-PT       *
C   GAUSSIAN-LENGENDRE INTEGRATION RULE NORMALIZED     *
C   TO THE INTERVAL [-1,1]                             *
C*******************************************************
          SUBROUTINE GRULE(N,X,W)
c      IMPLICIT REAL*8 (A-H,O-Z)
          DIMENSION X(N),W(N)
          M=(N+1)/2
          E1=N*(N+1)
          DO 1 I=1,M
               T=(4.0*I-1.0)*3.1415926536/(4*N+2)
               X0=(1.-(1.-1./N)/(8.*N*N))*COS(T)
               PKM1=1.0
               PK=X0
          DO 3 K=2,N
               T1=PK*X0
               PKP1=T1-PKM1-(T1-PKM1)/K+T1
               PKM1=PK
               PK=PKP1
3         CONTINUE
          DEN=1.-X0*X0
          D1=N*(PKM1-X0*PK)
          DPN=D1/DEN
          D2PN=(2.*X0*DPN-E1*PK)/DEN
          D3PN=(4.*X0*D2PN+(2.-E1)*DPN)/DEN
          D4PN=(6.*X0*D3PN+(6.-E1)*D2PN)/DEN
          U=PK/DPN
          V=D2PN/DPN
          H=-U*(1.+.5*U*(V+U*(V*V-U*D3PN/(3.*DPN))))
          P=PK+H*(DPN+0.5*H*(D2PN+H/3.*(D3PN+.25*H*D4PN)))
          DP=DPN+H*(D2PN+.5*H*(D3PN+H*D4PN/3.))
          H=H-P/DP
          X(I)=X0+H
          FX=D1-H*E1*(PK+.5*H*(DPN+H/3.*(D2PN+.25*H*(D3PN+.2*H*D4PN))))
          W(I)=2.*(1.-X(I)*X(I))/(FX*FX)
1         CONTINUE
          IF(M+M.GT.N)X(M)=0.
          RETURN
          END
```

```
C**********************************************
C           SUBROUTINE ARANGE                 *
C  THE SUBROUTINE  IS  TO ARANGE THE GAUSSIAN  *
C  QUA COORD. AND WEIGHTING FUNCTION           *
C**********************************************
      SUBROUTINE ARANGE(N,X,W)
c      IMPLICIT REAL*8 (A-H,O-Z)
      REAL X(N),W(N)
      M=N/2
      DO 1 I=1,M
           X(N-I+1)=-X(I)
           W(N-I+1)=W(I)
1     CONTINUE
      RETURN
      END
```